
TRYCATCH.TXT
OBJECT Table 60000 TempParam

{

  OBJECT-PROPERTIES

  {

    Date=08/07/17;

    Time=12:00:00;

    Modified=Yes;

    Version List=TRYCATCH;

  }

  PROPERTIES

  {

  }

  FIELDS

  {

    { 1   ;   ;Function Name       ;Text50         }

    { 2   ;   ;Param001            ;Text50         }

    { 3   ;   ;Param002            ;Text50         }

    { 4   ;   ;Param003            ;Text50         }

    { 5   ;   ;Param004            ;Text50         }

    { 6   ;   ;Param005            ;Text50         }

    { 7   ;   ;Param006            ;Text50         }

    { 8   ;   ;Param007            ;Text50         }

    { 9   ;   ;Param008            ;Text50         }

    { 10  ;   ;Param009            ;Text50         }

    { 11  ;   ;Param010            ;Text50         }

    { 12  ;   ;Param011            ;Text50         }

    { 13  ;   ;Param012            ;Text50         }

    { 14  ;   ;Param013            ;Text50         }

    { 15  ;   ;Param014            ;Text50         }

    { 16  ;   ;Param015            ;Text50         }

    { 17  ;   ;Param016            ;Text50         }

    { 18  ;   ;Param017            ;Text50         }

    { 19  ;   ;Param018            ;Text50         }

    { 20  ;   ;Param019            ;Text50         }

    { 21  ;   ;Param020            ;Text50         }

  }

  KEYS

  {

    {    ;Function Name                           ;Clustered=Yes }

  }

  FIELDGROUPS

  {

  }

  CODE

  {



    BEGIN

    END.

  }

}



OBJECT Codeunit 60000 Try/Catch

{

  OBJECT-PROPERTIES

  {

    Date=08/07/17;

    Time=12:00:00;

    Modified=Yes;

    Version List=TRYCATCH;

  }

  PROPERTIES

  {

    TableNo=60000;

    OnRun=BEGIN

            CASE "Function Name" OF

              'TryInsertItem':

                TryInsertItem(Rec);

              'TryPostPurchaseInvoice':

                TryPostPurchaseInvoice(Rec);

              // Add your Function Here

              'TryMyFunction':

                TryMyFunction(Rec);

              ELSE

                ERROR('The function ''%1'' is not handled by this Codeunit.', "Function Name");

            END;

          END;



  }

  CODE

  {



    LOCAL PROCEDURE TryInsertItem@1000000004(VAR _TempParam@1000000000 : TEMPORARY Record 60000);

    VAR

      Item@1000000002 : Record 27;

      ItemNo@1000000001 : Code[20];

      ItemDescription@1000000003 : Text[50];

      ItemUnitPrice@1000000004 : Decimal;

      AutomaticExtTexts@1000000005 : Boolean;

      BaseUnitOfMeasure@1000000006 : Code[10];

    BEGIN

      // Convert my generic parameters

      EVALUATE(ItemNo, _TempParam.Param001);

      EVALUATE(ItemDescription, _TempParam.Param002);

      EVALUATE(ItemUnitPrice, _TempParam.Param003);

      EVALUATE(AutomaticExtTexts, _TempParam.Param004);

      EVALUATE(BaseUnitOfMeasure, _TempParam.Param005);



      Item.INIT;

      Item.VALIDATE("No.", ItemNo);

      Item.VALIDATE(Description, ItemDescription);

      Item.VALIDATE("Unit Price", ItemUnitPrice);

      Item.VALIDATE("Automatic Ext. Texts", AutomaticExtTexts);

      Item.VALIDATE("Base Unit of Measure", _TempParam.Param005);

      Item.INSERT(TRUE);

    END;



    LOCAL PROCEDURE TryPostPurchaseInvoice@1000000000(VAR _TempParam@1000000000 : TEMPORARY Record 60000);

    VAR

      PurchaseHeader@1000000002 : Record 38;

      PurchPost@1000000003 : Codeunit 90;

      InvoiceNo@1000000001 : Code[20];

    BEGIN

      // Convert my generic parameters

      EVALUATE(InvoiceNo, _TempParam.Param001);



      PurchaseHeader.GET(PurchaseHeader."Document Type"::Invoice, InvoiceNo);

      PurchPost.RUN(PurchaseHeader);

    END;



    PROCEDURE TryMyFunction@1000000007(VAR _TempParam@1000000000 : TEMPORARY Record 60000);

    VAR

      MayParam001@1000000001 : Text[50];

      MayParam002@1000000002 : Integer;

      MayParam003@1000000003 : Decimal;

      MayParam004@1000000004 : Boolean;

      MayParam005@1000000005 : Date;

    BEGIN

      // This is a template function

      // Convert my generic parameters

      EVALUATE(MayParam001, _TempParam.Param001);

      EVALUATE(MayParam002, _TempParam.Param002);

      EVALUATE(MayParam003, _TempParam.Param003);

      EVALUATE(MayParam004, _TempParam.Param004);

      EVALUATE(MayParam005, _TempParam.Param005);



      // Do what you want with your paramaters

      ERROR('The parameters of ''%1'' are:\  %2\  %3\  %4\  %5\  %6',

        _TempParam."Function Name",

        MayParam001,

        MayParam002,

        MayParam003,

        MayParam004,

        MayParam005

      );

    END;



    BEGIN

    END.

  }

}



OBJECT Codeunit 60001 Try/Catch Example

{

  OBJECT-PROPERTIES

  {

    Date=09/07/17;

    Time=12:00:00;

    Modified=Yes;

    Version List=TRYCATCH;

  }

  PROPERTIES

  {

    OnRun=BEGIN

            Example1;

            // Example2;

            // Example3;

            // Example4;

          END;



  }

  CODE

  {



    PROCEDURE Example1@1000000002();

    VAR

      TempParam@1000000001 : TEMPORARY Record 60000;

      TryCatch@1000000000 : Codeunit 60000;

    BEGIN

      TempParam.DELETEALL;

      TempParam.INIT;

      TempParam."Function Name" := 'TryInsertItem';

      TempParam.Param001 := 'NAVNAB'; // No.

      TempParam.Param002 := 'NAV NAB BLOG'; // Description

      TempParam.Param003 := '150'; // Unit Price

      TempParam.Param004 := 'Yes'; // Automatic Ext. Texts

      TempParam.Param004 := 'Yes'; // Base Unit of Measure



      IF TryCatch.RUN(TempParam) THEN

        MESSAGE('OK')

      ELSE

        MESSAGE(GETLASTERRORTEXT);

    END;



    PROCEDURE Example2@1000000003();

    VAR

      TempParam@1000000001 : TEMPORARY Record 60000;

      TryCatch@1000000000 : Codeunit 60000;

    BEGIN

      TempParam.DELETEALL;

      TempParam.INIT;

      TempParam."Function Name" := 'TryPostPurchaseInvoice';

      TempParam.Param001 := 'NAVNAB'; // Invoice No.



      IF TryCatch.RUN(TempParam) THEN

        MESSAGE('OK')

      ELSE

        MESSAGE(GETLASTERRORTEXT);

    END;



    PROCEDURE Example3@1000000004();

    VAR

      TempParam@1000000001 : TEMPORARY Record 60000;

      TryCatch@1000000000 : Codeunit 60000;

    BEGIN

      TempParam.DELETEALL;

      TempParam.INIT;

      TempParam."Function Name" := 'TryMyFunction';

      TempParam.Param001 := 'Dummy Text'; // Text parameter

      TempParam.Param002 := '13'; // Integer parameter

      TempParam.Param003 := '13' + GetDecimalSeparator + '06'; // Decimal parameter

      TempParam.Param004 := 'No'; // Boolean parameter

      TempParam.Param005 := '01012999'; // Date parameter



      IF TryCatch.RUN(TempParam) THEN

        MESSAGE('OK')

      ELSE

        MESSAGE(GETLASTERRORTEXT);

    END;



    PROCEDURE Example4@1000000006();

    VAR

      TempParam@1000000001 : TEMPORARY Record 60000;

      TryCatch@1000000000 : Codeunit 60000;

    BEGIN

      TempParam.DELETEALL;

      TempParam.INIT;

      TempParam."Function Name" := 'TryMyFunction2';

      TempParam.Param001 := 'NAVNAB001';

      TempParam.Param002 := 'NAVNAB002';

      TempParam.Param003 := 'NAVNAB003';

      TempParam.Param004 := 'NAVNAB004';

      TempParam.Param005 := 'NAVNAB005';

      TempParam.Param006 := 'NAVNAB006';

      TempParam.Param007 := 'NAVNAB007';

      TempParam.Param008 := 'NAVNAB008';

      TempParam.Param009 := 'NAVNAB009';



      IF TryCatch.RUN(TempParam) THEN

        MESSAGE('OK')

      ELSE

        MESSAGE(GETLASTERRORTEXT);

    END;



    PROCEDURE GetDecimalSeparator@1000000001() : Text[1];

    VAR

      TempText@1000000000 : Text[30];

    BEGIN

      EXIT(COPYSTR(FORMAT(1/10), 2, 1));

    END;



    BEGIN

    END.

  }

}

